FONCTIONS D'UNE VARIABLE RÉELLE : DÉVELOPPEMENTS LIMITÉS ET DIFFÉRENTIELLES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Maple offre de riches possibilités pour les approximations mais il est notoirement pauvre dans le domaine des différentielles, qui ne sont pas traitées comme des objets mathématiques en soi. Les économistes, gros consommateurs de différentielles en microéconomie et en macroéconomie, toujours à l'affût pour mesurer l'impact d'une petite variation de la variable explicative sur la variable expliquée, doivent donc ruser pour les introduire dans leurs feuilles de travail. La dernière section de ce chapitre donne une méthode qui s'appuie sur les développements limités (en abrégé DL) et la récupération de leur partie principale.

Calculs de développements limités

Tout calcul de développement limité fait appel à la commande **series**. On donne 3 arguments : la fonction sous forme d'une expression, le point au voisinage duquel est faite l'approximation, enfin l'ordre du développement. Ce dernier n'est pas obligatoire dans la mesure où Maple adopte un développement à l'ordre 6 par défaut.

```
> restart;
  series(f(x),x=a);#développement en série (par défaut à l'ordre 6)d'une fonction quelconque
  series(f(x),x=a,3);#développement en série à l'ordre 3 demandé en option
```

$$f(a) + \mathrm{D}(f)(a)(x-a) + \frac{1}{2}\mathrm{D}^{(2)}(f)(a)(x-a)^2 + \frac{1}{6}\mathrm{D}^{(3)}(f)(a)(x-a)^3 + \frac{1}{24}\mathrm{D}^{(4)}(f)(a)(x-a)^4 + \frac{1}{120}\mathrm{D}^{(5)}(f)(a)(x-a)^5 + \mathrm{O}\left((x-a)^6\right)$$

$$f(a) + \mathrm{D}(f)(a)(x-a) + \frac{1}{2}\mathrm{D}^{(2)}(f)(a)(x-a)^2 + \mathrm{O}\left((x-a)^3\right)$$

On remarque que le reste est donné avec la notation de Landau. Le développement d'ordre n renvoyé par Maple correspond donc à un DL d'ordre $(n-1)$.

Que la fonction soit simple ou compliquée, Maple amène un résultat correct pourvu qu'elle soit suffisamment dérivable au point considéré :

```
> F:=sinh(x);
  G:=cos(x);
  series(F*G,x=0);#développement en série du produit des fonctions F et G au voisinage de 0
  series(F/G,x=0);#développement en série du quotient des fonctions F et G au voisinage de 0.
```

$$F := \sinh(x)$$

$$G := \cos(x)$$

$$x - \frac{1}{3}x^3 - \frac{1}{30}x^5 + \mathrm{O}\left(x^6\right) \tag{1.1}$$

$$x+\frac{2}{3}x^3+\frac{3}{10}x^5+O(x^6) \tag{1.1}$$

Récupération de la partie principale d'un développement

Pour extraire la partie principale d'un DL, on fait appel à la commande `convert(DL, polynom)`.

```
> restart;dl:=series(sin(x),x):prdl:=convert(dl,polynom);
```

$$prdl := x-\frac{1}{6}x^3+\frac{1}{120}x^5$$

On a alors la possibilité de tracer la fonction et son approximation polynomiale :

```
> plot({sin(x),prdl},x=-3/2*Pi..3/2*Pi);
```



Pour la recherche d'optima, les économistes s'intéressent surtout à la partie principale d'ordre 2 d'un DL. On procèdera en deux temps : demande d'un développement à l'ordre 3 puis extraction de la partie régulière. Par exemple :

```
> restart;dl:=series(x*log(x)-20*x,x=5,3):prdl:=convert(dl,
  polynom);
```

$$prdl := 5\ln(5)-100+(-19+\ln(5))(x-5)+\frac{1}{10}(x-5)^2$$

Différentielle

Commençons par quelques rappels sur la fonction différentielle.
Une fonction différentiable étant donnée, on a vu dans la section précédente que Maple donne aisément la partie principale de son DL(1) :

```
> restart;
  dl1:=series(f(x),x=a,2);#calcul du DL(1)
  prdl1:=convert(dl1,polynom);#récupération de la partie régulière
```

$$dl1 := f(a) + \mathrm{D}(f)(a)\,(x-a) + \mathrm{O}\left((x-a)^2\right)$$
$$prdl1 := f(a) + \mathrm{D}(f)(a)\,(x-a)$$

En retirant de cette expression la valeur prise par la fonction au point a, on obtient une valeur approchée de l'accroissement $f(x) - f(a)$, qui est, par définition, la différentielle de f en a. En économie, il est d'usage de remplacer l'accroissement $x - a$ par l'écriture compacte dx et, de la même manière, l'accroissement $f(x) - f(a)$ par dy.

```
> exp1:=prdl1-f(a);#écriture rigoureuse de la différentielle de f en a
  exp2:=subs(x-a=dx,%);#écriture usuelle de la différentielle de f en a
```

$$exp1 := \mathrm{D}(f)(a)\,(x-a)$$
$$exp2 := \mathrm{D}(f)(a)\,dx$$

Dès lors, l'application différentielle s'obtient par **unapply**. Son argument est noté **dx** conformément à l'usage.

```
> Dy:=unapply(exp2,dx);#création de l'application "différentielle de f en a"
```

$$Dy := dx \rightarrow \mathrm{D}(f)(a)\,dx$$

Traitons un exemple. On cherche la différentielle de la fonction $f(x) = \frac{\log(x)}{x}$ au point $x = 2$ puis la valeur prise par celle-ci en $dx = 0.01$.

```
> restart;
  f:=x->log(x)/x;#définition de la fonction
  prdl1:=convert(series(f(x),x=2,2),polynom);#partie principale du DL(1)en x=2
  exp1:=prdl1-f(2);#écriture rigoureuse de la différentielle
  exp2:=subs(x-2=dx,%);#écriture usuelle de la différentielle
  Dy:=unapply(exp2,dx);#création de la fonction différentielle de f en x=2
  evalf(Dy(0.01));#valeur prise par la différentielle en 0.01
```

$$f := x \rightarrow \frac{\log(x)}{x}$$
$$prdl1 := \frac{1}{2}\ln(2) + \left(\frac{1}{4} - \frac{1}{4}\ln(2)\right)(x-2)$$
$$exp1 := \left(\frac{1}{4} - \frac{1}{4}\ln(2)\right)(x-2)$$
$$exp2 := \left(\frac{1}{4} - \frac{1}{4}\ln(2)\right)dx$$

$$Dy := dx \rightarrow \left(\frac{1}{4} - \frac{1}{4}\ln(2)\right) dx$$

$$0.000767132048$$

Faire plus simple est plus que souhaitable. En anticipant sur les fonctions de plusieurs variables, on crée une fonction de trois variables dont la première est la fonction mathématique considérée, la seconde le point où est calculé la dérivée de la fonction et la troisième l'accroissement de la variable indépendante.

```
> restart;
  Dy:=(f,a,dx)->D(f)(a)*dx;#la fonction-procédure Dy crée la
  différentielle de la fonction f au point a
```

$$Dy := (f, a, dx) \rightarrow \mathrm{D}(f)(a)\, dx$$

Dans le cas général où `f`, `a` et `dx` sont quelconques, on aura :

```
> Dy(f,a,dx);
```

$$\mathrm{D}(f)(a)\, dx$$

Et dans le cas particulier de l'exemple précédent :

```
> f:=x->log(x)/x;#on reprend la fonction de l'exemple
  précédent
  Dy(f,2,dx);#différentielle de f en 2
  Dy(f,2,0.01);#calcul de la différentielle de f en 2 pour un
  accroissement de 0.01 de la variable indépendante
  evalf(Dy(f,2,0.01));#évaluation du résultat précédent sous
  forme d'un nombre réel
```

$$f := x \rightarrow \frac{\log(x)}{x}$$

$$\left(\frac{1}{4} - \frac{1}{4}\ln(2)\right) dx$$

$$0.002500000000 - 0.002500000000\ln(2)$$

$$0.000767132048$$

Exercices

Exercice M.1

Calculer le développement limité au voisinage du point a et, si possible, au voisinage de 0, des fonctions suivantes à l'ordre demandé :

1. $f(x) = e^x$ à l'ordre 5
2. $f(x) = \ln x$ à l'ordre 5
3. $f(x) = \ln(1 + x)$ à l'ordre 5 et à l'ordre 1
4. $f(x) = \dfrac{1}{1+x}$ à l'ordre 5 puis à l'ordre 1
5. $f(x) = \dfrac{1}{1-x}$ à l'ordre 5 puis à l'ordre 1
6. $f(x) = (1 + x)^{\alpha}$ à l'ordre 5 puis à l'ordre 1
7. $f(x) = \cos x$ à l'ordre 5
8. $f(x) = \sin x$ à l'ordre 5 puis à l'ordre 1
9. $f(x) = \mathrm{tg}\, x$ à l'ordre 5

Exercice M.2

En économie, on érige l'approximation $\ln(1+h) \approx h$ en quasi-dogme. Ainsi, le logarithme d'un indice d'une grandeur statistique serait automatiquement son taux de croissance. Montrez graphiquement qu'il s'agit bien d'une approximation et non d'une égalité.

Exercice E.1

Appliquez la fonction-procédure de calcul d'une différentielle exposée dans la section **Différentielle** aux situations microéconomiques suivantes.

1. La fonction d'utilité générique $U = U(c)$, puis la fonction d'utilité spécifique $U = \frac{c^{\alpha}}{\alpha - 1}$.

2. La fonction de production générique à une variable (le travail) $Q = F(L)$, puis la fonction de production particulière $Q = L^{\frac{2}{3}}$.

3. La fonction de coût total générique $C = C(Q)$, puis la fonction de coût particulière $C = 3Q^3 + 5Q^2 - 10Q + 25$.